

HE
18.5
.A37
no.
DOT-
TSC-
UMTA-
75-21
v.3

PB 254 675

NO. UMTA-MA-06-0025-75-16.IXI

Dept. of Transportation

JUL 13 1976

Library

AN ASSESSMENT OF THE CRASHWORTHINESS
OF EXISTING URBAN RAIL VEHICLES
Volume III: Train-Collision Model
Users Manual

D. J. Segal



NOVEMBER 1975
FINAL REPORT

DOCUMENT IS AVAILABLE TO THE PUBLIC
THROUGH THE NATIONAL TECHNICAL
INFORMATION SERVICE, SPRINGFIELD,
VIRGINIA 22161

Prepared for
U.S. DEPARTMENT OF TRANSPORTATION
URBAN MASS TRANSPORTATION ADMINISTRATION
Office of Research and Development
Washington DC 20590

NOTICE

This document is disseminated under the sponsorship of the Department of Transportation in the interest of information exchange. The United States Government assumes no liability for its contents or use thereof.

NOTICE

The United States Government does not endorse products or manufacturers. Trade or manufacturers' names appear herein solely because they are considered essential to the object of this report.

1. Report No. UMTA-MA-06-0025-75-16.III		2. Government Accession No.		3. Recipient's Catalog No.	
4. Title and Subtitle AN ASSESSMENT OF THE CRASHWORTHINESS OF EXISTING URBAN RAIL VEHICLES. Volume III: Train-Collision Model Users Manual				5. Report Date November 1975	
7. Author(s) D. J. Segal				6. Performing Organization Code	
9. Performing Organization Name and Address Calspan Corporation* P.O. Box 235 Buffalo NY 14221				8. Performing Organization Report No. DOT-TSC-UMTA-75-21.III v.3	
12. Sponsoring Agency Name and Address U.S. Department of Transportation Urban Mass Transportation Administration, Office of Research and Development Washington DC 20590				10. Work Unit No. (TRAIS) UM604/R6745	
				11. Contract or Grant No. DOT-TSC-681	
				13. Type of Report and Period Covered Final Report August 1973 - June 1975	
15. Supplementary Notes *Under contract to:				U.S. Department of Transportation Transportation Systems Center Kendall Square Cambridge MA 02142	
16. Abstract The crashworthiness of existing urban rail vehicles (passenger cars) and the feasibility of improvements in this area were investigated by Calspan Corporation under contract to the Transportation Systems Center of the U.S. Department of Transportation in its role as systems manager for the Urban Mass Transportation Administration's Urban Rail Supporting Technology Program. Both rail-car structural configurations and impact absorption devices were studied. From this work, recommendations for engineering standards for urban rail vehicles will be developed. This final report issued under the crashworthiness effort covers: <ol style="list-style-type: none"> 1. The development of analytical tools to predict passenger threat - environment during collision. 2. Criteria for predicting passenger injury due to train collisions. 3. An application of injury criteria and analytic models to predict passenger injuries resulting from collisions of trains that represent existing construction types. 4. A preliminary investigation of applying impact absorption devices to transit vehicles. 5. A design study of car structural configurations for improved impact energy management. 6. A review of engineering standards for Urban Rail Car Crashworthiness. The report consists of three volumes: Volume I: Analyses and Assessments of Vehicles, Chapters 1 Through 7 Volume II: Analyses and Assessments of Vehicles, Chapters 8 Through 12, Appendixes and References Volume III: Train Collision Model Users Manual.					
17. Key Words Rail Car Crashworthiness; Train Collision, Transportation Safety; Crashworthiness, Mathematical Models			18. Distribution Statement DOCUMENT IS AVAILABLE TO THE PUBLIC THROUGH THE NATIONAL TECHNICAL INFORMATION SERVICE, SPRINGFIELD, VIRGINIA 22161		
19. Security Classif. (of this report) Unclassified		20. Security Classif. (of this page) Unclassified		21. No. of Pages 66	
				22. Price	


PREFACE

This Train-Collision Model Users Manual presents documentation of a simplified mathematical model and computer program of the longitudinal dynamics of rail cars involved in collisions and an assessment of the range of subsequent injuries sustained by passengers. The Train-Collision Model, was originally developed in the support of a program of assessment of the crashworthiness of existing urban rail vehicles by the Calspan Corporation for the U.S. Department of Transportation, Transportation Systems Center under Contract No. DOT-TSC-681. Further improvements in the model and documentation were made as reported herein under that contract.

METRIC CONVERSION FACTORS

Approximate Conversions to Metric Measures

Symbol	When You Know	Multiply by	To Find	Symbol
LENGTH				
in	inches	2.5	centimeters	cm
ft	feet	30	meters	m
yd	yards	0.9	kilometers	km
mi	miles	1.6		
AREA				
in ²	square inches	6.5	square centimeters	cm ²
ft ²	square feet	0.09	square meters	m ²
yd ²	square yards	0.8	square kilometers	km ²
mi ²	square miles	2.6	hectares	ha
MASS (weight)				
oz	ounces	28	grams	g
lb	pounds	0.45	kilograms	kg
	short tons (2000 lb)	0.9	tonnes	t
VOLUME				
tsp	teaspoons	5	milliliters	ml
Tbsp	tablespoons	15	milliliters	ml
fl oz	fluid ounces	30	milliliters	ml
c	cups	0.24	liters	l
pt	pints	0.47	liters	l
qt	quarts	0.95	liters	l
gal	gallons	3.8	liters	l
ft ³	cubic feet	0.03	cubic meters	m ³
yd ³	cubic yards	0.76	cubic meters	m ³
TEMPERATURE (exact)				
°F	Fahrenheit temperature	5/9 (after subtracting 32)	Celsius temperature	°C



Approximate Conversions from Metric Measures

Symbol	When You Know	Multiply by	To Find	Symbol
LENGTH				
mm	millimeters	0.04	inches	in
cm	centimeters	0.4	inches	in
m	meters	3.3	feet	ft
m	meters	1.1	yards	yd
km	kilometers	0.6	miles	mi
AREA				
cm ²	square centimeters	0.16	square inches	in ²
m ²	square meters	1.2	square yards	yd ²
km ²	square kilometers	0.4	square miles	mi ²
ha	hectares (10,000 m ²)	2.5	acres	
MASS (weight)				
g	grams	0.035	ounces	oz
kg	kilograms	2.2	pounds	lb
t	tonnes (1000 kg)	1.1	short tons	
VOLUME				
ml	milliliters	0.03	fluid ounces	fl oz
l	liters	2.1	pints	pt
l	liters	1.06	quarts	qt
l	liters	0.26	gallons	gal
m ³	cubic meters	35	cubic feet	ft ³
m ³	cubic meters	1.3	cubic yards	yd ³
TEMPERATURE (exact)				
°C	Celsius temperature	9/5 (then add 32)	Fahrenheit temperature	°F




TABLE OF CONTENTS

<u>Section</u>	<u>Page</u>
1. INTRODUCTION	1-1
2. MATHEMATICAL-MODEL DESCRIPTION	2-1
2.1 Rail-Car Model	2-1
2.2 Calculation of Forces	2-5
2.3 Rail-Car Integration	2-7
2.4 Passenger-Injury Assessment	2-9
3. TRAIN-COLLISION PROGRAM	3-1
3.1 Input	3-1
3.2 Output	3-6
3.3 Sample Runs	3-9
3.4 Program Glossary	3-18
3.5 Program Listing	3-20
3.6 Program Flowcharts	3-31

LIST OF FIGURES

<u>Figure</u>		<u>Page</u>
2-1	Simplified Schematic Diagram of Typical Mode 1 Train Configuration	2-2
2-2	Simplified Schematic Diagram of Typical Mode 2 Train Configuration	2-4
2-3	Force-Deflection Characteristic	2-5
2-4	Force-Calculation Logic	2-6
2-5	Improved Euler Method Integration	2-8
3-1	Train-Collision Program Input	3-2
3-2	Sample Listing of Program Inputs	3-7
3-3	Sample Program Output	3-8
3-4	Inputs For Sample Run 1	3-10
3-5	Mass Displacement - Sample Run 1	3-11
3-6	Inputs For Sample Run 2	3-12
3-7	Car Displacements - Sample Run 2	3-13
3-8	Inputs For Sample Run 3	3-14
3-9	Mass Displacement--Sample Run 3	3-15
3-10	Inputs For Sample Run 4	3-16
3-11	Car Deflections - Sample Run 4	3-17

1. INTRODUCTION

The Train Collision Model is a simplified dynamic simulation of rail-car collisions with automatic estimation of the degree of severity sustained by passengers of each car in a train during second collisions with various types of interior surfaces at different initial distances. The rail cars are modeled as rigid bodies separated by nonlinear, energy-absorbing force deflection characteristics with the first car of a train impacting a rigid barrier. Two modes of operation are possible, the first assuming that an individual force-deflection characteristic acts between each car. For this mode, the deflection of each car (except the first) is assumed to be one-half the relative displacement between the two adjacent car centers of gravity. The second mode of operation allows the individual deflections of each end of each car to be obtained through the use of an inertial mass between the springs representing the car ends. This mode is useful in approximating the different deformations of cars that over-/underride.

Passenger injury-severity distributions are approximated by determining the relative velocity between a mass representing a passenger and the car at times at which the passenger has moved a number of input distances through the car. The relative velocities are subsequently used to calculate a severity index assuming that a passenger in that situation strikes a car fixed object that nulls the relative velocity in a number of input distances at a constant force level.

It must be recognized that the mathematical model and computer simulation described in this report is a very simplified approximation of an extremely complex phenomenon. In this respect, care should be exercised in interpretation of model results as they apply to the real world. While it is expected that generalized trends in overall rail-car crash situations may be properly obtained, results in an absolute sense for specific crash situations should not be indiscriminantly interpreted.

2. MATHEMATICAL-MODEL DESCRIPTION

2.1 RAIL-CAR MODEL

A simplified schematic diagram of the mode 1 rail car model is shown in Figure 2-1. The model consists of a barrier with optional movement at a constant velocity, and a number of rail cars (maximum of 20) idealized as rigid bodies separated by nonlinear, energy-absorbing springs. A coordinate system is fixed at the center of gravity of each car and passenger at the instant of impact and by inspection, the acceleration of a car i relative to that coordinate system is given by:

$$\ddot{X}_{c_i} = \frac{1}{M_i} (F_{c_{i+1}} - F_{c_i}) - a_{c_0} g,$$

where a_{c_0} is an arbitrary input deceleration (because of, for example, train braking).

The forces acting between cars arise from deformation of the car ends. Tabular force-deflection characteristics are input to the model, and forces are generated based on the relative positions and velocities of adjoining cars. In this mode of operation, it is assumed that the deflection of the nonlinear springs is shared equally between cars although two force-deflection tables are input so that different characteristics may act between different cars. Thus, for the purpose of determining the force acting between cars i and $i + 1$, the deflection and deflection velocity are

$$\delta_{i+1} = \frac{1}{2} (X_{c_{i+1}} - X_{c_i}),$$

$$\dot{\delta}_{i+1} = \frac{1}{2} (\dot{X}_{c_{i+1}} - \dot{X}_{c_i}).$$




Figure 2-1 Simplified Schematic Diagram of Typical Mode 1 Train Configuration

For the case of the first car of the train,

$$\delta_1 = X_{C1} - X_w ,$$

$$\dot{\delta}_1 = \dot{X}_{C1} - V_w .$$

Knowing the deflections and relative velocities of the cars, the forces are determined as described in Section 2.2.

A simplified schematic of the mode 2 rail car model is shown in Figure 2-2. This mode of operation allows independent specification of the force-deflection properties of each adjacent car. An inertial mass is used to separate the cars so that the deflection of each car is calculated independently. This mode is useful in determining the deflections of adjacent car ends when override occurs. In this mode, every other mass is a representation of a rail car, thus a maximum of 10 cars can be simulated. The deflection and deflection velocity of each of the nonlinear springs are given by:


$$\delta_{i+1} = X_{Ci+1} - X_{Ci} ,$$

$$\dot{\delta}_{i+1} = \dot{X}_{Ci+1} - \dot{X}_{Ci} ,$$

and, for the first spring,

$$\delta_1 = X_{C1} - X_w ,$$

$$\dot{\delta}_1 = \dot{X}_{C1} - V_w .$$



M_i = WEIGHT/g, $i = 1, 3, 5, \dots$

M_j = (WFACTR)*WEIGHT/g, $j = 2, 4, 6, \dots$

* p = PASSENGER (OCCUPANT)

Figure 2-2 Simplified Schematic Diagram of Typical Mode 2 Train Configuration

2.2 CALCULATION OF FORCES

The calculation of forces acting between rail cars is based on the known deflection and time rate of change of deflection. A general table of force-deflection characteristics is used to interpolate the force level when the deflection is increasing. Unloading and reloading after unloading has occurred takes place along a straight line with a slope of URATE as illustrated in Figure 2-3. If reloading back to F_m , δ_m occurs, calculation of the force along the original curve is resumed.




Figure 2-3 Force-Deflection Characteristic

A conceptual flow diagram of the force calculation logic is shown in Figure 2-4.




Figure 2-4. Force-Calculation Logic

2.3 RAIL-CAR INTEGRATION

Integration of the differential equations of motion of the rail cars is by the Improved Euler Method as illustrated by Figure 2-5. The procedure is as follows:

- a. At the beginning of a time increment, $t = t_m$, determine the acceleration of each car, $a_m^{(1)}$, by calculation of the forces.
- b. Assuming a constant acceleration, $a_m^{(1)}$, over the time increment, Δt , approximate the velocity and position of the car at the end of the interval ($t = t_m + \Delta t = t_{m+1}$) by:

$$v_{m+1}^{(0)} = v_m + a_m^{(1)} \Delta t,$$

$$x_{m+1}^{(0)} = x_m + v_m \Delta t + \frac{1}{2} a_m^{(1)} \Delta t^2.$$

- c. Determine the acceleration at the end of the interval based on the estimated position and velocity, $a_{m+1}^{(0)}$, by calculation of forces.
- d. Assume that the acceleration over the interval is the average of the above calculated accelerations, $\bar{a}_m = 1/2 (a_m^{(1)} + a_{m+1}^{(0)})$, and compute the revised position and velocity at the end of the interval:

$$\begin{aligned} v_{m+1}^{(1)} &= v_m + \bar{a}_m \Delta t \\ &= v_m + \frac{1}{2} (a_m^{(1)} + a_{m+1}^{(0)}) \Delta t, \end{aligned}$$

$$\begin{aligned} x_{m+1}^{(1)} &= x_m + v_m \Delta t + \frac{1}{2} \bar{a}_m \Delta t^2 \\ &= x_m + v_m \Delta t + \frac{1}{4} (a_m^{(1)} + a_{m+1}^{(0)}) \Delta t^2. \end{aligned}$$




Figure 2-5 Improved Euler Method Integration

2.4 PASSENGER-INJURY ASSESSMENT

The severity of a given crash situation depends in a general sense on the deceleration level acting on the occupant. The deceleration level is a function of the type of surface struck and the relative velocity at impact. Rail cars are, in general, a relatively uncontrolled environment for the occupants after an impact and thus a deterministic approach to modeling occupant dynamics is not at this time warranted.

The approach taken to obtain a general assessment of rail-car crash severity in this model was to determine a range of severity based on the distance the passenger travels before impact and the crush distance of the struck surface. One passenger is simulated in each car and the position and velocity of the passenger are determined assuming that an input constant deceleration acts (simulating for instance, sliding friction). When the position of the passenger relative to the car reaches any one of a number of input spacings (the distance the passenger travels before striking an object), the relative velocity is used to determine an injury-severity index. It is further assumed that the passenger decelerates to a relative velocity of zero in a number of input stopping distances (crush distances) at a constant deceleration level. Note that the effects of ridedown (car deceleration during passenger impact) are neglected in this simplified evaluation.

Given the above assumptions, and assuming a passenger relative velocity of V_r at impact, and a deceleration distance of d , the passenger deceleration level (in g-units) is

$$a_p = \frac{V_r^2}{2dg} \cdot$$

The severity index is:

$$SI = \int_t a_p^{2.5} dt = a_p^{2.5}(t),$$

or

$$SI = a_p^{2.5} \frac{v_r}{a_p g} = a_p^{1.5} \frac{v}{g} .$$

3. TRAIN-COLLISION PROGRAM

3.1 INPUT

Refer to Figure 3-1.

Card 1 Format 20A4

TITLE(I), I = 1, 20 Run title. Up to 80 characters of
alphanumeric information describing
the run

Card 2 Format 4F10.0, I5

TO	initial-simulation time	sec
TF	final-simulation time	sec
DT	integration-time increment	sec
DTP	output-print interval	sec
MODE	MODE = 0 assumes series springs - between cars	
	MODE = 1 assumes inertial masses acting between cars	

Card 3 Format I5, 5X, 5F10.0

NCARS	number of cars in the train (if MODE = 1, total number of masses)	
WEIGHT	weight of each car	lb
VW	barrier velocity	mph
VTRN	train initial velocity	mph
ACO	train-braking deceleration level	g
APO	passenger deceleration	g

TRAIN COLLISION MODEL	
INPUT CODING FORM	
1 of 2	
1	TITLE(I), I = 1, 20
2	TO TF DT DTP MODE
3	NCARS WEIGHT VW VTRN ACO APO
4	SPAC(I), I = 1, 6
5	GR(I), I = 1, 5
6	FCO(I, 1), I = 1, 8
6A	FCO(9, 1) FCO(10, 1)
7	FCO(I, 2), I = 1, 8
7A	FCO(9, 2) FCO(10, 2)
8	DEL(I, 1), I = 1, 8
8A	DEL(9, 1) DEL(10, 1)

TRAIN COLLISION MODEL		2 of 2
INPUT CODING FORM		
CARD	1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57 58 59 60	
9	DEL(1, 2), I = 1, 8	
9A	DEL(9, 2) DEL(10, 2)	
10	URATE(1) URATE(2)	
11	IF(I), I = 1, NCARS	
12	WFACTR	

Figure 3-1 Train-Collision Program Input, Sheet 2 of 2

Card 4 Format 6F10.0

SPAC(I), I = 1,6

passenger-spacing distances

before impact

ft

Card 5 Format 6F10.0

CR(I), I = 1,5

passenger-deceleration distances

after impact

in.

Card 6 Format 8G10.0

FCO(I,1), I = 1,8

force table for first force-

deflection characteristic

1b

Card 6A Format 2G10.0

FCD(I,1), I = 9,10

Card 7 Format 8G10.0

FCO(I,1), I = 1,8

force table for second force-

deflection characteristic

1b

Card 7A Format 2G10.0

FCO(I,2), I = 9,10

Card 8 Format 8G10.0

DEL(I,1), I = 1,8

deflection table for first force-

deflection characteristic

ft

Card 8A Format 2G10.0

DEL(I,1), I = 9,10

Card 9 Format 8G10.0

DEL(I,2), I = 1,8

deflection table for second

Card 9A Format 2G10.0

force-deflection characteristic ft

DEL(I,2), I = 9,10

Card 10 Format 2G10.0

URATE(1)

unloading rates for first and

URATE(2)

second force-deflection

characteristics

lb/ft

Card 11 Format 20I4

IF(I), I = 1, NCARS

force-deflection characteristic

number (1 or 2) to be used for

spring I. Note that spring I

acts between masses I-1 and I

Card 12 Format 1G10.0 (to be read only if MODE = 1)

WFACTR

multiplier used to reduce the

weight of masses acting between

cars if MODE = 1

0 WFACTR 1.0.

3.2 OUTPUT

Output from the Train-Collision Program includes a listing of program inputs as shown in Figure 3-2, and a time history of dynamic output as shown in Figure 3-3. The dynamic output consists of:

- a. The simulated time, in sec
- b. The car (or spring) number (note that car number 0 indicates the barrier for which only the position and velocity are output) and for MODE = 1, even-numbered cars represent inertial masses
- c. XC, the displacement of each car relative to its position at impact, in ft
- d. DEFL, the deflection of the indicated spring.
Spring I acts between masses I-1 and I. In the case of MODE = 0, the deflection of spring I is $1/2 (X_{c_i} - X_{c_{i-1}})$. In the case of MODE = 1, the deflection of spring I is $(X_{c_i} - X_{c_{i-1}})$
- e. VC, the velocity of the indicated mass in ft/sec
- f. AC, the acceleration of the indicated mass, in g
- g. XPC, the relative displacement of the passenger with respect to the car, for the indicated car, in ft
- h. VPC, the relative velocity of the passenger with respect to the car for the indicated car, in ft/sec
- i. F, the force produced by the indicated spring, in lb.

TRAIN COLLISION MODEL
SAMPLE RUN NO. 1 SPRING - MASS SYSTEM

TC	=	0.0	SEC	NO. OF CARS	=	1
TF	=	0.100	SEC	CARWEIGHT	=	10.0 LBS
DT	=	0.0020	SEC	BARRIER VELOCITY	=	0.0 MPH
DT PRINT	=	0.0020	SEC	TRAIN VELOCITY	=	30.00 MPH
MUDE	=	C		TRAIN DECEL.	=	0.0 G+5
				PASSENGER DECEL.	=	0.0 G+5

TRAIN FORCE-DEFLECTION CHARACTERISTICS		
FCO(1)	DEL(1)	FCO(2)
LBS	FT.	LBS
DEL(2)	FT.	

[illegible]

```
UNLOADING RATES: URATE(1)=      1226.60LB/FT
                   URATE(2)=      1226.60LB/FT
```

CAR NO.	FORCE-DEFLECTION CHARACTERISTIC NO.
1	1
1	1

PASSENGER SPACING DISTANCE - FT	PASSENGER DECEL. DISTANCE - IN
0.10	1.00
0.20	2.00
0.30	3.00
0.40	4.00
0.50	5.00
0.60	

Figure 3-2 Sample Listing of Program Inputs

TIME SEC	CAR NO.	XC FT.	DEFL FT.	VC FT/SEC	AC G'S	XPC FT.	VPC FT/SEC	F L8.
0.0	0	0.0		0.0				
	1	0.0	0.0	44.00	0.0	0.0	0.0	
0.002	0	0.0		0.0				
	1	0.09	0.09	43.65	-10.79	0.00	0.35	.1C789E+03
0.004	0	0.0		0.0				
	1	0.17	0.17	42.62	-21.36	0.00	1.38	.21365E+03
0.006	0	0.0		0.0				
	1	0.26	0.26	40.91	-31.60	0.01	3.09	.31605E+03
0.008	0	0.0		0.0				
	1	0.34	0.34	38.57	-41.35	0.02	5.43	.41347E+03
0.010	0	0.0		0.0				
	1	0.41	0.41	35.61	-50.44	0.03	8.39	.50439E+03
0.012	0	0.0		0.0				
	1	0.48	0.48	32.10	-58.74	0.05	11.90	.56737E+03
0.014	0	0.0		0.0				
	1	0.54	0.54	28.08	-66.11	0.08	15.92	.66111E+03
0.016	0	0.0		0.0				
	1	0.59	0.59	23.62	-72.44	0.11	20.38	.72445E+03
0.018	0	0.0		0.0				
	1	0.63	0.63	18.79	-77.64	0.16	25.21	.77639E+03
0.020	0	0.0		0.0				
	1	0.67	0.67	13.66	-81.61	0.21	30.34	.81612E+03
0.022	0	0.0		0.0				
	1	0.69	0.69	8.32	-84.30	0.28	35.68	.84301E+03
0.024	0	0.0		0.0				
	1	0.70	0.70	2.84	-85.66	0.36	41.16	.85665E+03
0.026	0	0.0		0.0				
	1	0.70	0.70	-2.67	-85.66	0.45	46.67	.85659E+03
0.028	0	0.0		0.0				
	1	0.69	0.69	-8.15	-84.33	0.54	52.15	.84328E+03
0.030	0	0.0		0.0				
	1	0.67	0.67	-13.49	-81.67	0.65	57.49	.81670E+03
0.032	0	0.0		0.0				
	1	0.63	0.63	-18.62	-77.73	0.77	62.62	.77727E+03
0.034	0	0.0		0.0				
	1	0.59	0.59	-23.46	-72.56	0.90	67.46	.72563E+03

Figure 3-3 Sample Program Output

3.3 SAMPLE RUNS

Four sample runs using the Train-Collision Program are presented in this section. The first is a simple linear spring-mass system used to check the accuracy of the integrator. As shown in the program inputs in Figure 3-4, the mass weighs 10 lb and the spring has a rate of 1226 lb/ft. The exact solution to this system yields a natural frequency of 10 cps, and with an initial velocity of 30 mph, the maximum displacement is 0.7003 ft, and maximum acceleration is -85.85 g. The results from the program indicate a maximum displacement of 0.70 ft (Figure 3-5) and maximum acceleration of -85.85 g.

The second and third sample runs compare program results under different modes of operation. Inputs for the second run are shown in Figure 3-6. Two masses, each of which weighs 10 lb, with linear springs, of rates 1226 lb/ft impact a stationary barrier at 30 mph. Displacements of the masses are shown in Figure 3-7.

The third sample run duplicates the conditions of the second (Figure 3-8) except that program mode 1 is used. The two 10-lb masses are separated by a mass of 0.1 lb (WFACTR = 0.01). Results of this run are shown in Figure 3-9. Note that the displacements of the two 10-lb masses are nearly identical with the second run. Although this mode of operation may be useful in determining deflections of individual cars with different force-deflection characteristics, care must be exercised in the choice of integration step size to insure that stability in the integration of the small mass dynamics is maintained.

The fourth sample run is typical of the inputs required for a four-car train. The inputs used are shown in Figure 3-10 and the deflections of the forward ends of each car are shown in Figure 3-11.

SAMPLE RUN NO. 1 TRAIN COLLISION MODEL SPRING - MASS SYSTEM					
TC =	0.0	SEC	NO. OF CARS =	1	
TF =	0.000	SEC	CAR WEIGHT =	10.0	LBS
DT =	0.00100	SEC	BARRIER VELOCITY =	0.0	MPH
UT PRINT =	0.0010	SEC	TRAIN VELOCITY =	30.00	MPH
MODE =	0		TRAIN DECEL. =	0.0	G/S
			PASSENGER DECEL. =	0.0	G/S
TRAIN FORCE-DEFLECTION CHARACTERISTICS					
FCO(1)	DEL(1)	FCO(2)	DEL(2)		
LBS	FT.	LBS	FT.		
0.0	0.0	0.0	0.0		
1226.0	1.00	1226.0	1.00		
12260.0	10.00	12260.0	10.00		
0.0	0.0	0.0	0.0		
0.0	0.0	0.0	0.0		
0.0	0.0	0.0	0.0		
0.0	0.0	0.0	0.0		
0.0	0.0	0.0	0.0		
0.0	0.0	0.0	0.0		
0.0	0.0	0.0	0.0		
UNLOADING RATES: URATE(1)= 1226.00LB/FT URATE(2)= 1226.00LB/FT					
CAR NO.					
FORCE-DEFLECTION CHARACTERISTIC NU.	1				
PASSENGER SPACING	PASSENGER DECEL.				
DISTANCE - FT	DISTANCE - IN				
0.10	1.00				
0.20	2.00				
0.30	3.00				
0.40	4.00				
0.50	5.00				
0.60					




Figure 3-5 Mass Displacement - Sample Run 1

TRAIN COLLISION MODEL									
SAMPLE RUN NO. 2 TWO MASS SYSTEM MODE 1									
TO =	0.0	SEC	NO. OF CARS =	2					10.0 LBS
TF =	0.1000	SEC	CARWEIGHT =						0.0 MPH
DT =	0.00200	SEC	BARRIER VELOCITY =						30.00 MPH
DT PRINT =	0.0020	SEC	TRAIN VELOCITY =						0.0 G'S
MODE =	0		TRAIN DECEL. =						0.0 G'S
			PASSENGER DECEL. =						
TRAIN FORCE-DEFLECTION CHARACTERISTICS									
FCO(1)	DEL(1)	FCO(2)	DEL(2)						
LBS	FT.	LBS	FT.						
0.0	0.0	0.0	0.0						
1226.0	1.00	1226.0	1.00						
12260.0	10.00	12260.0	10.00						
0.0	0.0	0.0	0.0						
0.0	0.0	0.0	0.0						
0.0	0.0	0.0	0.0						
0.0	0.0	0.0	0.0						
0.0	0.0	0.0	0.0						
0.0	0.0	0.0	0.0						
0.0	0.0	0.0	0.0						
0.0	0.0	0.0	0.0						
UNLOADING RATES: URATE(1)=				1226.00LBS/FT	URATE(2)= 1226.00LBS/FT				
CAR NO.									
FORCE-DEFLECTION CHARACTERISTIC NO.	1	2							
	1	1							
PASSENGER SPACING									
DISTANCE - FT									
0.10	1.00								
0.20	2.00								
0.30	3.00								
0.40	4.00								
0.50	5.00								
0.60									

Figure 3-6 Inputs For Sample Run 2




Figure 3-7 Car Displacements - Sample Run 2

TRAIN COLLISION MODEL
SAMPLE RUN NO. 5 3 MASS MODE 1

TC = 0.0 SEC NO. OF CARS = 3
 TF = 0.1000 SEC CARRIAGE WEIGHT = 10.0 LBS
 DT = 0.00200 SEC BARRIER VELOCITY = 0.0 MPH
 DT PRINT = 0.0020 SEC TRAIN VELOCITY = 30.00 MPH
 MODE = 1 TRAIN DECEL. = 0.0 G'S
 PASSENGER DECEL. = 0.0 G'S

TRAIN FORCE-DEFLECTION CHARACTERISTICS

FCO(1) LBS	DEL(1) FT.	FCO(2) LBS	DEL(2) FT.
0.0	0.0	0.0	0.0
1226.0	1.00	1226.0	1.00
12260.0	10.00	12260.0	10.00
0.0	0.0	0.0	0.0
0.0	0.0	0.0	0.0
0.0	0.0	0.0	0.0
0.0	0.0	0.0	0.0
0.0	0.0	0.0	0.0
0.0	0.0	0.0	0.0
0.0	0.0	0.0	0.0
0.0	0.0	0.0	0.0
0.0	0.0	0.0	0.0

UNLOADING RATES: URATE(1) = 1226.00LB/FT URATE(2) = 1226.00LB/FT

CAR NO.
FORCE-DEFLECTION CHARACTERISTIC NO. 1 2 3

PASSENGER SPACING
DISTANCE - FT

PASSENGER DECEL.
DISTANCE - IN

0.10	1.00
0.20	2.00
0.30	3.00
0.40	4.00
0.50	5.00
0.60	

WFACTOR = 0.0100

Figure 3-8 Inputs For Sample Run 3




Figure 3-9 Mass Displacement--Sample Run 3

TRAIN COLLISION MODEL
SAMPLE RUN NO. 4 4 CAR TRAIN

TV = 0.0 SEC NO. OF CARS = 4
 TF = 0.5000 SEC CARWEIGHT = 60000.0 LBS
 DT = 0.00100 SEC BARRIER VELOCITY = 0.0 MPH
 DT PRINT = 0.0100 SEC TRAIN VELOCITY = 20.00 MPH
 MODE = 0 TRAIN DECEL. = 0.0 G'S
 PASSENGER DECEL. = 0.0 G'S

TRAIN FORCE-DEFLECTION CHARACTERISTICS

FC0(1) LBS	DEL(1) FT.	FC0(2) LBS	DEL(2) FT.
0.0	0.0	0.0	0.0
165000.0	0.25	165000.0	0.25
0.0	0.26	0.0	0.26
0.0	0.42	0.0	0.42
720000.0	0.50	720000.0	0.50
720000.0	4.58	720000.0	4.58
1170000.0	4.67	1170000.0	4.67
1170000.0	66.67	1170000.0	66.67
1170000.0	70.00	1170000.0	70.00
1170000.0	75.00	1170000.0	75.00

UNLOADING RATES: URATE(1) = 8640000.00LB/FT

URATE(2) = 8640000.00LB/FT

CAR NO.
FORCE-DEFLECTION CHARACTERISTIC NO. 1 2 3 4

PASSENGER SPACING
DISTANCE - FT
2.00
4.00
6.00
9.00
12.00
24.00

PASSENGER DECEL.
DISTANCE - IN
0.25
1.00
2.00
4.00
12.00

Figure 3-10 Inputs For Sample Run 4




Figure 3-11 Car Deflections - Sample Run 4

3.4 PROGRAM GLOSSARY

<u>Program Symbol</u>	<u>Definition</u>
AC	car or mass acceleration
ACO	car-braking deceleration
AIMP	passenger impact deceleration
AP	passenger acceleration
APO	passenger-braking deceleration
CR	passenger crush distance
DEF	car deflection
DEFV	car-deflection velocity
DEL	deflection table
DM	deflection start of unloading
DP	previous value of DEF (at T-DT)
DR	zero force intercept of reloading curve
DT	integration increment
DTP	printout interval
DVP	previous value of DEJV (at T-DT)
FC	intermass force
FCO	force table
FM	force level at start of unloading
G	acceleration of gravity, 32.2 ft/sec ²
IF	force-deflection number indicator
IMP	passenger-impact indicator
MODE	indicator for simulation mode of operation

<u>Program Symbol</u>	<u>Definition</u>
N, NCARS	number of cars or masses
SI	passenger severity index table
SPAC	passenger spacing before impact
T	simulated time
TF	end time
TITLE	run title
TO	initial time
URATE	unloading rate
VC	car or mass velocity
VIMP	passenger relative velocity at impact
VP	passenger velocity
VTRN	train initial velocity
VW	barrier velocity
WEIGHT	car weight
WFACTR	factor for reducing weight of inter-car inertial masses
WGT	car or mass weight
XC	car position
XP	passenger position
XW	barrier position

3.5 PROGRAM LISTING

```

C      ***  RAIL CAR COLLISION MODEL  ****
C****
COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XP(20),VP(20),
1      AP(20),VIMP(20,6),DVP(20),DP(20),DM(20),FM(20),URATE(2),
2      DR(20),DEL(10,2),FCO(10,2),IF(20),TITLE(20),TO,NCARS,TF,
3      WEIGHT,DT,VW,DTP,VTRN,MODE,ACO,APO,SPAC(6),CR(5),WFACTR,
4      XW ,T,G,N,AIMP(20,6,5),SI(20,6,5)
C****
C      CLEAR COMMON STORAGE TO ZERO
C****
1 CALL CLEAR(IMP(1),SI(20,6,5))
C****
C      INITIALIZE OCCUPANT CONTACT INDICATOR
C****
4 DO 2 I=1,20
2 IMP(I) = 1
C****
C      OBTAIN INPUT AND INITIALIZE
C****
3 CALL INPUT
TPRNT = TO+DTP
G = 32.2
FPS = 5280./3600.
VWALL = VW *FPS
VTRN = VTRN*FPS
N = NCARS
N1 = 1
DO 31 I=1,N
31 WGT(I) = WEIGHT
C****
C      MODE=1 INDICATES THAT EVERY OTHER CAR IS AN INERTIAL MASS
C      OF WEIGHT WFACTR*WGT BETWEEN CARS SO THAT DIFFERENT
C      FORCE-DEFLECTION CHARACTERISTICS CAN BE USED ON
C      ADJACENT CARS
C****
C****
C      INITIALIZATION
C****
IF(MODE.NE.1) GO TO 28
DO 29 I=2,N,2
29 WGT(I) = WGT(I)*WFACTR
28 XW = 0.0
VW = VWALL
T = TO
DO 32 I=1,N
FC(I) = 0.0
XC(I) = 0.0
VC(I) = VTRN
AC(I) = ACO*G
XP(I) = 0.0
VP(I) = VTRN
32 AP(I) = 0.0
CALL OUTPUT(1)
C****
C      BEGIN INTEGRATION LOOP.  INTEGRATION BY MODIFIED EULER METHOD

```

```

C****
  10 XW = VW*T
C****
C    EVALUATE ACCELERATION AT T
C****
      CALL FORCE
      DO 707 I=1,N
      ACT = 0.0
      IF(ABS(VC(I)).GT.0.01) ACT = ACO*VC(I)/ABS(VC(I))
      XDD = (FC(I+1)-FC(I))*G/WGT(I)+ACT
C****
C    PROJECT POSITION AND VELOCITY AT T+DT
C****
      XD = VC(I)+XDD*DT
      X = XC(I)+VC(I)*DT+.5*XDD*DT*DT
      XC(I) = X
      VC(I) = XD
      707 AC(I) = XDD
      T = T+DT
      XW = VW*T
C****
C    EVALUATE ACCELERATION AT T+DT
C****
      CALL FORCE
      DO 706 I=1,N
      ACT = 0.0
      IF(ABS(VC(I)).GT.0.01) ACT = ACO*VC(I)/ABS(VC(I))
      XDD = (FC(I+1)-FC(I))*G/WGT(I)+ACT
C****
C    REVISE POSITION AND VELOCITY AT T+DT
C****
      XD1 = VC(I)-AC(I)*DT
      XD = XD1+.5*(AC(I)+XDD)*DT
      X1 = XC(I)-XD1*DT-.5*AC(I)*DT*DT
      X = X1+XD1*DT+.25*(AC(I)+XDD)*DT*DT
      XC(I) = X
      VC(I) = XD
      706 AC(I) = XDD
      N1 = 1
      IF(MODE.EQ.1) N1 = 2
C****
C    INTEGRATE POSITION AND VELOCITY OF PASSENGERS IN EACH CAR
C****
      DO 40 I=1,N,N1
      VDIF = VC(I)-VP(I)
      AP(I) = 0.0
      IF(VDIF.NE.0.0) AP(I) = APO*VDIF/ABS(VDIF)
      VP(I) = VP(I)+AP(I)*DT
      XP(I) = XP(I)+VP(I)*DT+.5*AP(I)*DT*DT
C****
C    CALCULATE RELATIVE POSITION AND VELOCITY OF PASSENGER
C****
      XPR = XP(I)-XC(I)
      VPR = VP(I)-VC(I)
      IS = IMP(I)

```

```

      IF(IS.GE.7) GO TO 40
C****
C      IF PASSENGER HAS MOVED INPUT DISTANCES OF SPAC RELATIVE TO
C      CAR, LOAD RELATIVE VELOCITY INTO VIMP FOR SUBSEQUENT SEVERITY
C      INDEX CALCULATIONS
C      SPAC MUST BE IN INCREASING ORDER
C****
      IF(XPR.LT.SPAC(IS)) GO TO 40
      IMP(I) = IS+1
      VIMP(I,IS) = VPR
40    CONTINUE
      IF(T.GT.TF) GO TO 900
C****
C      OUTPUT IF PRINT INTERVAL HAS BEEN REACHED
C****
      IF(TPRNT.GT.T+.1*DT) GO TO 10
      CALL OUTPUT(2)
      TPRNT = TPRNT+DTP
      GO TO 10
C****
C      AT END OF RUN , CALCULATE SEVERITY INDEX
C****
900  CALL SIC
      GO TO 4
      END

```

```

      SUBROUTINE OUTPUT(IND)
C****
C      PRINTS DYNAMIC OUTPUT FROM TRAIN COLLISION MODEL
C****
      COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XP(20),VP(20),
1         AP(20),VIMP(20,6),DVP(20),DP(20),DM(20),FM(20),URATE(2),
2         DR(20),DEL(10,2),FCO(10,2),IF(20),TITLE(20),TO,NCARS,TF,
3         WEIGHT,DT,VW,DTP,VTRN,MODE,ACO,APO,SPAC(6),CR(5),WFACTR,
4         XW      ,T,G,N,AIMP(20,6,5),SI(20,6,5)
C****
C      FOR IND=1, CALCULATE IPRNT FOR OUTPUT PAGING
C****
      IF(IND.GT.1) GO TO 10
      IPRNT = 55/(NCARS+2)
5     LL = 0
C****
C      OUTPUT PAGE HEADINGS
C****
      WRITE(6,2000)
2000  FORMAT(1H1,1X,4HTIME,6X,3HCAR,5X,2HXC,8X,4HDEFL,6X,2HVC,
1         8X,2HAC,8X,3HXP,7X,3HVPC,7X,1HF /
2         2X,3HSEC,7X,3HNO.,5X,3HFT.,7X,3HFT.,7X,6HFT/SEC,4X,3HG'S,
3         7X,3HFT.,7X,6HFT/SEC,4X,3HLB. // )
10    LL = LL+1
C****
C      LL IS A COUNTER FOR PAGING
C****
      IF(LL.GT.IPRNT) GO TO 5
      I = 0
C****
C      OUTPUT FOR BARRIER
C****
      WRITE(6,2001) T,I,XW,VW
2001  FORMAT(1H0,F8.3,2X,I2,2X,F8.2,12X,F8.2)
      FCTR = .5
      IF(MODE.EQ.1) FCTR = 1.0
      DO 34 I= 1,N
      FI = FC(I)
      XI = XC(I)
      VI = VC(I)
      AI = AC(I)/G
      XPC = XP(I)-XI
      VPC = VP(I)-VI
      IF(I.GT.1) GO TO 32
      DI = XI-XW
      GO TO 31
32    DI = (XI-XC(I-1))*FCTR
C****
C      OUTPUT DYNAMIC DATA FOR MASSES AND SPRINGS
C****
      31 WRITE(6,2002) I,XI,DI,VI,AI,XPC,VPC,FI
2002  FORMAT(1H ,10X,I2,6(2X,F8.2),2X,E10.5)
      34 CONTINUE
      RETURN
      END

```



```

SUBROUTINE FORCE
C****
C      CALCULATES FORCE ACTING BETWEEN CARS AS A FUNCTION OF DISTANCE
C****
COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XP(20),VP(20),
1      AP(20),VIMP(20,6),DVP(20),DP(20),DM(20),FM(20),URATE(2),
2      DR(20),DEL(10,2),FCO(10,2),IF(20),TITLE(20),TG,NCARS,TF,
3      WEIGHT,DT,VW,DTP,VTRN,MODE,ACO,APO,SPAC(6),CR(5),WFACTR,
4      XW ,T,G,N,AIMP(20,6,5),SI(20,6,5)
DO 50 I=1,N
C****
C      NF INDICATES THE APPROPRIATE FORCE-DEFLECTION CHARACTERISTIC
C      FOR SPRING I
C****
NF = IF(I)
IF(I.NE.1) GO TO 51
DEF = XC(1)-XW
DEFV = VC(1)-VW
GO TO 52
51 FACTOR = 1.0
C****
C      IF MODE=0, FORCE DEFLECTION CHARACTERISTICS BETWEEN CARS
C      ARE SERIES COMBINATIONS OF THE INDIVIDUAL CAR PROPERTIES.
C      THE DEFLECTION OF EACH CAR IS ASSUMED TO BE .5*THE TOTAL.
C****
IF(MODE.EQ.0) FACTOR = .5
DEF = (XC(I)-XC(I-1))*FACTOR
DEFV = VC(I)-VC(I-1)
52 IF(DEF.GT.0.) GO TO 10
F = 0.0
GO TO 300
10 IF(DEFV.GT.0.) GO TO 100
200 IF(DVP(I).LT.0.) GO TO 220
C****
C      SET UNLOADING CURVE
C****
DM(I) = DEF
FM(I) = FC(I)
C****
C      CALCULATE UNLOADING FORCE
C****
220 F = FM(I)-URATE(NF)*(DM(I)-DEF)
IF(F.GE.0.0) GO TO 300
230 F = 0.0
GO TO 300
100 IF(DVP(I).LE.0.0) GO TO 150
110 IF(DR(I).EQ.0.0) GO TO 180
GO TO 160
C****
C      SET RELOADING CURVE
C      DR IS THE ZERO FORCE INTERCEPT FOR RELOADING
C****
150 DR(I) = DP(I)-FC(I)/URATE(NF)
C****
C      CALCULATE FORCE FOR RELOADING ALONG UNLOADING CURVE

```

```

C****
160 FP = URATE(NF)*(DEF-DR(I))
    IF(FP.LT.0.0) FP = 0.0
C****
C      INTERPOLATE FORCE ON LOADING CURVE
C****
180 DO 185 K=2,10
    IF(DEF.GT.DEL(K,NF)) GO TO 185
    K1 = K-1
    F = FC0(K1,NF)+(FC0(K,NF)-FC0(K1,NF))*(DEF-DEL(K1,NF))
1    / (DEL(K,NF)-DEL(K1,NF))
    GO TO 186
185 CONTINUE
    F = FC0(9,NF)+(FC0(10,NF)-FC0(9,NF))*(DEF-DEL(9,NF))/
1    (DEL(10,NF)-DEL(9,NF))
186 CONTINUE
    IF(DR(I).EQ.0.0) GO TO 300
C****
C      IF RELOADING, MAKE SURE FORCE LEVEL ON RELOADING CURVE DOES NOT
C      EXCEED FORCE ON ORIGINAL LOADING TABLE
C****
    IF(F.GT.FP) GO TO 190
    DR(I) = 0.0
    GO TO 300
190 F = FP
300 FC(I) = F
C****
C      SET PREVIOUS VALUES FOR NEXT TIME
C****
    DP(I) = DEF
    DVP(I) = DEFV
50 CONTINUE
    FC(N+1) = 0.0
    RETURN
    END

```

SUBROUTINE SIC

```

C****
C      CALCULATES SEVERITY INDEX MATRIX FOR EACH CAR FOR PASSENGER
C      DISTANCES OF SPAC BEFORE IMPACT AND STOPPING DISTANCES
C      OF CR. NOTE: THAT THE RELATIVE VELOCITY OF THE PASSENGER
C      W/TO THE CAR AT SPACING DISTANCES SPAC ARE VIMP AS CALCULATED
C      IN MAIN.
C***
COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XP(20),VP(20),
1      AP(20),VIMP(20,6),DVP(20),DP(20),DM(20),FM(20),URATE(2),
2      DR(20),DEL(10,2),FCO(10,2),IF(20),TITLE(20),TO,NCARS,TF,
3      WEIGHT,UT,VW,DTP,VTRN,MODE,ACO,APO,SPAC(6),CR(5),WFACTR,
4      XW,T,G,N,AIMP(20,6,5),SI(20,6,5)
N1 = 1
IF(MODE.EQ.1) N1 = 2
DO 800 I=1,N,N1
DO 800 K=1,6
DO 800 M=1,5
C****
C      AIMP IS CONSTANT ACCELERATION FROM VELOCITY VIMP IN DISTANCE CR.
C****
AIMP(I,K,M) = G.*VIMP(I,K)**2/(CR(M)*G)
C****
C      SEVERITY INDEX IS INTEGRAL OF ACCELERATION**2.5 OVER TIME
C****
SI(I,K,M) = AIMP(I,K,M)**1.5*VIMP(I,K)/G
800 CONTINUE
C****
C      OUTPUT SEVERITY INDEX TABULATION AS A FUNCTION OF PASSENGER
C      SPACING AND CRUSH DISTANCE FOR EACH CAR
C****
WRITE(6,3000)
3000 FORMAT(1H1, // 40X, 14HSEVERITY INDEX )
DO 950 I=1,N,N1
WRITE(6,3001) I, (SPAC(K), K=1,6)
3001 FORMAT(1H0, 10X, 3HCAR, 13 / 3X, 5HCRUSH, 22X, 12HSPACING - FT /
1      4X, 3HIN., 3X, 6F10.2 )
DO 950 K=1,5
WRITE(6,3002) CR(K), (SI(I,J,K), J=1,6)
3002 FORMAT(1H0, F7.2, 2X, 6F10.1 / (1X, F7.2, 2X, 6F10.1) )
950 CONTINUE
IF(MODE.EQ.1) WRITE(6,3003)
3003 FORMAT(1H0, // 47H NOTE: MODE=1, THEREFORE PASSENGERS ASSOCIATED
1      20HWITH EVERY OTHER CAR )
RETURN
END

```

SUBROUTINE INPUT

```

C****
C      READS TRAIN COLLISION MODEL INPUT FROM CARDS AND PRINTS INPUT DATA
C****
COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XP(20),VP(20),
1      AP(20),VIMP(20,6),DVP(20),DP(20),DM(20),FM(20),URATE(2),
2      DR(20),DEL(10,2),FCO(10,2),IF(20),TITLE(20),TO,NCARS,TF,
3      WEIGHT,DT,VW,DTP,VTRN,MODE,ACO,APO,SPAC(6),CR(5),WFACTR,
4      XW      ,T,G,N,AIMP(20,6,5),SI(20,6,5)
C****
C      READ INPUT
C****
C****
C      RUN TITLE CARD, 80 CHARACTERS
C****
      READ(5,99) (TITLE(I),I=1,20)
99      FORMAT(20A4)
C****
C      PROGRAM CONTROL
C****
      READ(5,98) TO,TF,DT,DTP,MODE
98      FORMAT(4F10.0,I5)
      READ(5,97) NCARS,WEIGHT,VW,VTRN,ACO,APO
97      FORMAT(I5,5X,5F10.0)
      READ(5,96) (SPAC(I),I=1,6)
      READ(5,96) (CR(I),I=1,5)
96      FORMAT(6F10.0)
C****
C      EACH FCO AND DEL REQUIRES 2 CARDS WITH 2 VALUES ON SECOND CARD
C      BLANK CARD MUST BE FURNISHED IF NO DATA FOR LAST TWO ENTRIES.
C****
      READ(5,95) (FCO(I,1),I=1,10)
      READ(5,95) (FCO(I,2),I=1,10)
      READ(5,95) (DEL(I,1),I=1,10)
      READ(5,95) (DEL(I,2),I=1,10)
      READ(5,95) URATE(1),URATE(2)
95      FORMAT(8G10.0)
C****
C      IF INDICATES WHICH FCO-DEL TABLE USED FOR EACH SPRING. MUST
C      BE 1 OR 2
C****
      READ(5,94) (IF(I),I=1,NCARS)
94      FORMAT(20I4)
C****
C      WFACTR READ ONLY IF MODE = 1
C****
      IF(MODE.EQ.1) READ(5,95) WFACTR
C****
C      PRINT INPUT
C****
      WRITE(6,1000) (TITLE(I),I=1,20)
1000      FORMAT(1H1,40X,21HTRAIN COLLISION MODEL / 20X,20A4)
      WRITE(6,1001) TO,NCARS,TF,WEIGHT,DT,VW,DTP,VTRN,MODE,ACO,APO
      WRITE(6,1002) (FCO(I,1),DEL(I,1),FCO(I,2),DEL(I,2),I=1,10)
      WRITE(6,1003) URATE(1),URATE(2)

```

```

WRITE(6,1004) (I,I=1,NCARS)
WRITE(6,1005) (IF(I),I=1,NCARS)
WRITE(6,1006) (SPAC(I),CR(I),I=1,5),SPAC(6)
IF(MODE.EQ.1) WRITE(6,1007) WFACTR
1007 FORMAT(1H0, // 9H WFACTR =,F10.4)
1001 FORMAT(1H0,25X,4HTO =,F8.4,4H SEC,8X,13HNO. OF CARS =,14, /
1 26X,4HTF =,F8.4,4H SEC,9X,11HCARWEIGHT =,F12.1,4H LBS /
2 26X,4HDT =,F8.5,4H SEC,3X,18HBARRIER VELOCITY =,F12.2,4H MPH/
3 20X,10HDT PRINT =,F8.4,4H SEC,5X,16HTRAIN VELOCITY =,F12.2,
4 4H MPH /
5 24X,6HMODE =,14,15X,14HTRAIN DECEL. =,F12.2,4H G'S /
6 45X,18HPASSENGER DECEL. =,F12.3,4H G'S )
1002 FORMAT(1H0, // 10X,38HTRAIN FORCE-DEFLECTION CHARACTERISTICS /
1 13X,6HFCO(1),10X,6HDEL(1),10X,6HFCO(2),10X,6HDEL(2) /
2 14X,3HLBS,13X,3HFT.,13X,3HLBS,13X,3HFT. //
3 (10X,F10.1,8X,F7.2,7X,F10.1,8X,F7.2) )
1003 FORMAT(1H0,10X,26HUNLOADING RATES: URATE(1)=,F14.2,5HLB/FT,
2 10X,9HURATE(2)=,F14.2,5HLB/FT /)
1004 FORMAT(1H0, // 5X,7HCAR NO.,28X,15I5 )
1005 FORMAT(1H ,4X,35HFORCE-DEFLECTION CHARACTERISTIC NO. ,15I5 )
1006 FORMAT(1H0, // 10X,17HPASSENGER SPACING,13X,16HPASSENGER DECEL. /
1 13X,13HDISTANCE - FT,17X,13HDISTANCE - IN /
2 (13X,F8.2,22X,F8.2) )
RETURN
END

```

```


SUBROUTINE CLEAR(A,B)
C   CLEARS (SETS TO ZERO) A BLOCK OF STORAGE IDENTIFIED BY
C   THE ADDRESSES OF THE TWO ARGUMENTS
C   CALL CLEAR(P,Q) WILL CAUSE ALL BYTES TO BE SET TO
C   ZERO FROM ADDRESS P THROUGH THE FULL-WORD ADDRESS AT Q
  DIMENSION A(1),B(1)
  B(1) = 1.0
  I = 0
10 IF(B(1).EQ.0.0) RETURN
  I = I+1
  A(I) = 0.0
  GO TO 10
END

```

3.6 PROGRAM FLOWCHARTS


CHART TITLE - PROCEDURES

*** RAIL CAR COLLISION MODEL *****



[illegible]

CHART TITLE - PROCEDURES




J4/11/74

AUTOFLOW CHART SET ~~454~~ TRAIN COLLISION MODEL

CHART TITLE - NON-PROCEDURAL STATEMENTS

```
COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XP(20),VP(20),  
AP(20),VIMP(20,6),DVP(20),OP(20),OM(20),FM(20),URATE(2),  
OR(20),OEL(10,2),FCO(10,2),IF(20),TITLE(20),TO,NCARS,TF,  
WEIGHT,OT,VW,OTP,VTRN,MOOE,ACO,APD,SPAC(6),CR(5),WFACTR,  
XW      ,T,G,N,AIMP(20,6,5),SI(20,6,5)
```

CHART TITLE - SUBROUTINE OUTPUT(INO)



09/11/74

AUTOFLOW CHART SET -- TRAIN COLLISION MODEL

CHART TITLE - NON-PROCEOURAL STATEMENTS

```
COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XP(20),VP(20),
      AP(20),VIMP(20,6),DVP(20),OP(20),OM(20),FMI(20),URATE(2),
      CR(20),OEL(10,2),FCO(10,2),IF(20),TITLE(20),TO,NCARS,TF,
      WEIGHT,OT,VW,OTP,VTRN,MOOE,ACO,APO,SPAC(6),CR(5),WFACTR,
      XM      ,T,G,N,AIMP(20,6,5),S1(20,6,5)
2000  FORMAT(1H1,1X,4HTIME,6X,3HCAR,5X,2HXC,8X,4HDEFL,6X,2HVC,
      8X,2HAC,8X,3HXPC,7X,3HVPC,7X,1HF      /
      2X,3HSEC,7X,3HNO,5X,3HFT,7X,3HFT,7X,6HFT/SEC,4X,3HG'S,
      7X,3HFT,7X,6HFT/SEC,4X,3HL8.      // )
2001  FORMAT(1H0,F8.3,2X,I2,2X,F8.2,12X,F8.2)
2002  FORMAT(1H ,10X,I2,6(2X,F8.2),2X,E10.5)
```

CHART TITLE - SUBROUTINE FORCE

/ FORCE /

02,22*-->*

CALCULATES FORCE
ACTING BETWEEN CARS
AS A FUNCTION OF
DISTANCE

```

      *
      * NDT: 01
      *
      * BEGIN OD LOOP
      * 50 I = 1, N
      *
      *
  
```

09,17-->|

NF INDICATES THE
APPROPRIATE
FORCE-DEFLECTION
CHARACTERISTIC
FOR SPRING 1






CHART TITLE - SUBROUTINE FORCE



09/11/74


AUTOFLOW CHART SET

TRAIN COLLISION MODEL

CHART TITLE - NON-PROCEEDURAL STATEMENTS

COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XP(20),VP(20),
AP(20),VIMP(20,6),DVP(20),OP(20),OM(20),FM(20),URATE(2),
OR(20),OEL(10,2),FCO(10,2),IF(20),TITLE(20),TO,NCARS,TF,
WEIGHT,OT,VW,OTP,VTRN,MODE,ACO,APO,SPAC(6),CR(5),WFACTR,
XW ,T,G,N,AIMP(20,6,5),SI(20,6,5)

CHART TITLE - SUBROUTINE SIC



09/11/74

AUTOFLOW CHART SET -- TRAIN COLLISION MOOEL

CHART TITLE - NON-PROCEDURAL STATEMENTS

```
COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XPI(20),VPI(20),
      AP(20),VIMP(20,6),DVP(20),OP(20),OM(20),FM(20),URATE(2),
      OR(20),OEL(10,2),FCO(10,2),IF(20),TITLE(20),TO,NCARS,TF,
      WEIGHT,OT,VH,OTP,VTRN,MOOE,ACO,APD,SPAC(6),CR(5),WFACTR,
      XM ,T,G,N,AIMP(20,6,5),SI(20,6,5)
3000  FORMAT(1H1, // 40X,14HSEVERITY INOE X )
3001  FORMAT(1H0,10X,3HCAR,13 / 3X,5HCRUSH,22X,12HSPACING - FT /
      4X,3MIN.,3X,6F10.2 )
3002  FORMAT(1H0,F7.2,2X,6F10.1 / (1X,F7.2,2X,6F10.1) )
3003  FORMAT(1H0, // 4TH NOTE: MOOE=1, THEREFORE PASSENGERS ASSOCIATED
      20HWITH EVERY OTHER CAR )
```

CHART TITLE - SUBROUTINE INPUT





CHART TITLE - SUBROUTINE INPUT



09/11/74

AUTOFLOW CHART SET GALE TRAIN COLLISION MODEL

CHART TITLE - NON-PROCEDURAL STATEMENTS

```

COMMON IMP(20),WGT(20),FC(21),XC(20),VC(20),AC(20),XP(20),VP(20),
      AP(20),VIMP(20,6),DVP(20),OP(20),OM(20),FM(20),URATE(2),
      OR(20),OEL(10,2),FCO(10,2),IF(20),TITLE(20),T0,NCARS,TF,
      WEIGHT,OT,VW,OTP,VTRN,MODE,ACO,APD,SPAC(6),CR(5),WFACTR,
      XN ,T,G,N,ALMP(20,6,5),SI(20,6,5)

99  FORMAT(20A4)
98  FORMAT(4F10.0,I5)
97  FORMAT(15,5X,5F10.0)
96  FORMAT(6F10.0)
95  FORMAT(8G10.0)
94  FORMAT(20I4)

1000 FORMAT(1H1,40X,21HTRAIN COLLISION MODEL / 20X,20A4)
1007 FORMAT(1H0, // 9H WFACTR =,F10.4)
1001 FORMAT(1H0,25X,4HTD =,F8.4,4H SEC,8X,13HNO. OF CARS =,I4, /
      26X,4HTF =,F8.4,4H SEC,9X,11HCARWEIGHT =,F12.1,4H LBS /
      26X,4HDT =,F8.5,4H SEC,3X,18HBARRIER VELOCITY =,F12.2,4H MPH/
      20X,10HDT PRINT =,F8.4,4H SEC,5X,16HTRAIN VELOCITY =,F12.2,
      4H MPH /
      24X,6HMODE =,I4,15X,14HTRAIN DECEL. =,F12.2,4H G'S /
      45X,18HPASSENGER DECEL. =,F12.3,4H G'S )

1002 FORMAT(1H0, // 10X,38HTRAIN FORCE-DEFLECTION CHARACTERISTICS /
      13X,6HFCO(1),10X,6HDEL(1),10X,6HFCO(2),10X,6HDEL(2) /
      14X,3HLBS,13X,3HFT.,13X,3HLBS,13X,3HFT. //
      (10X,F10.1,8X,F7.2,7X,F10.1,8X,F7.2) )

1003 FORMAT(1H0,10X,26HUNLOADING RATES: URATE(1)=,F14.2,5HLB/FT,
      10X,9HURATE(2)=,F14.2,5HLB/FT /)

1004 FORMAT(1H0, // 5X,7HCAR NO.,28X,15I5 )
1005 FORMAT(1H ,4X,35HFORCE-DEFLECTION CHARACTERISTIC NO. ,15I5 )
1006 FORMAT(1H0, // 10X,17HPASSENGER SPACING,13X,16HPASSENGER DECEL. /
      10X,13H01STANCE - FT,17X,13H01STANCE - IN /
      (13X,F8.2,22X,F8.2) )


```

09/11/74

AUTOFLOW CHART SET

TRAIN COLLISION MODEL

CHART TITLE - SUBROUTINE CLEAR(A,B)



09/11/74

AUTOFLOW CHART SET

--

TRAIN COLLISION MODEL

CHART TITLE - NON-PROCEDURAL STATEMENTS

DIMENSION A(1),B(1)



HE 18.5 .A37
no. DOT-TSC-
UMTA-
75-21

v.3

BORROWER

Detty, C.I.
Baker, P

Form DOT F 172

DOT LIBRARY



00009378